



Using SAS Formats

WUSS 2022 / Paper 146-2022

Tom Kari, Tom Kari Consulting Ltd.

Formats Basics



SAS formats are used to group and transform results

If we submit:

```
/* Tabulate price by cylinders */  
proc tabulate data=sashelp.cars;  
    class MSRP cylinders;  
    keylabel n="Cars";  
    table MSRP, cylinders;  
  
run;
```


We can create useful results with a format:

```
proc format;
    value MSRPFmt
        low <- 10000="Less than $10,000"
        10000 <- 20000="$10,000 to $20,000"
        20000 <- 30000="$20,000 to $30,000"
        30000 <- 40000="$30,000 to $40,000"
        40000 <- 50000="$40,000 to $50,000"
        50000 <- 60000="$50,000 to $60,000"
        60000 - high="Greater than $60,000";
run;
```

```
/* Tabulate price by cylinders */
proc tabulate data=sashelp.cars;
    format MSRP MSRPFmt.;
    class MSRP cylinders;
    keylabel n="Cars";
    table MSRP, cylinders;
run;
```

To produce:

	Cylinders						
	3	4	5	6	8	10	12
	Cars	Cars	Cars	Cars	Cars	Cars	Cars
MSRP							
\$10,000 to \$20,000	1	87	.	10	.	.	.
\$20,000 to \$30,000	.	36	.	93	9	.	.
\$30,000 to \$40,000	.	9	5	58	16	.	.
\$40,000 to \$50,000	.	4	2	19	24	1	.
\$50,000 to \$60,000	.	.	.	5	15	.	.
Greater than \$60,000	.	.	.	5	23	1	3

This is the easiest and fastest way to create and use formats, with no lasting effect:

- If you don't specify otherwise, the format is created in the "WORK" library;
- The "WORK" library is searched for formats by default;
- At the end of your SAS session, everything in the "WORK" library is deleted.

Permanent formats



If you want to set up a format that can be retained after the end of your SAS session, it's a little more complicated. You'll need:

- As before, a name for the format;
- An assigned permanent SAS library;
- A bit of additional work to get all of the pieces working together.

Step 1: A permanent SAS library

```
/* Define a permanent SAS library */  
libname CARlib "/home/xxx/wuss";
```

```
70  
71      /* Define a permanent SAS library */  
72      libname CARlib "/home/xxx/wuss";  
NOTE: Libref CARLIB was successfully assigned as follows:  
      Engine:          V9  
      Physical Name: /home/xxx/wuss  
73  
--
```

Step 2: Create the format

```
proc format library=CARlib;
  value MSRPfmt
    low -< 10000="Less than $10,000"
    10000 -< 20000="$10,000 to $20,000"
    20000 -< 30000="$20,000 to $30,000"
    30000 -< 40000="$30,000 to $40,000"
    40000 -< 50000="$40,000 to $50,000"
    50000 -< 60000="$50,000 to $60,000"
    60000 - high="Greater than $60,000";
run;
```

```
70
71      proc format library=CARlib;
72
72      ! value MSRPfmt
73      low -< 10000="Less than $10,000"
74      10000 -< 20000="$10,000 to $20,000"
75      20000 -< 30000="$20,000 to $30,000"
76      30000 -< 40000="$30,000 to $40,000"
77      40000 -< 50000="$40,000 to $50,000"
78      50000 -< 60000="$50,000 to $60,000"
79      60000 - high="Greater than $60,000";
NOTE: Format MSRPfmt has been written to CARLIB.FORMATS.
80      run;
```

NOTE: PROCEDURE FORMAT used (Total process time):


How formats are stored

SAS Library named CARlib:

Is this directory

Contains this file
(a SAS catalog)

Contains this format



Step 3: Tell SAS where to look for your formats

```
options fmtsearch=(CARlib);  
run;
```

Exploring formats



What libraries do I have allocated?

```
libname _all_ list;  
run;
```

```
71          libname _all_ list;  
NOTE: Libref=   WORK  
       Scope=   IOM ROOT COMP ENV  
       Engine=   V9  
       Access=   TEMP  
       Physical Name= /saswork/SAS_work58630000D137_odaws04-us  
       Filename= /saswork/SAS_work58630000D137_odaws04-usw2.od  
       Inode Number= 2186403  
       Access Permission= rwx-----  
       Owner Name=  
       File Size=                4KB  
       File Size (bytes)= 4096  
NOTE: Libref=   CARLIB  
       Scope=   IOM ROOT COMP ENV  
       Engine=   V9  
       Physical Name= /home/          /wuss  
       Filename= /home,          /wuss  
       Inode Number= 2151633962  
       Access Permission= rwxr-xr-x  
       Owner Name=  
       File Size=                0KB  
       File Size (bytes)= 90  
NOTE: Libref=   WEBWORK  
       Scope=   IOM ROOT COMP ENV  
       Engine=   V9  
       Physical Name= /home/          /.sasstudio/webwork/RS6624  
       Filename= /home/          /.sasstudio/webwork/RS6624  
       Inode Number= 2374916192
```

What about a single library?

```
libname CARlib list;  
run;
```

```
70  
71      libname CARlib list;  
NOTE: Libref=  CARLIB  
      Scope=    IOM ROOT COMP ENV  
      Engine=   V9  
      Physical Name= /home/⟨      ⟩ ./wuss  
      Filename= /home/⟨      ⟩ i/wuss  
      Inode Number= 2151633962  
      Access Permission= rwxr-xr-x  
      Owner Name=  ⟨      ⟩  
      File Size=           0KB  
      File Size (bytes)= 90  
72      run;  
--
```

What's in my library?

```
proc datasets lib=CARlib nolist;
           contents data=_all_;
quit;
```

The DATASETS Procedure

Directory	
Libref	CARLIB
Engine	V9
Physical Name	/home/c /uss
Filename	/hor
Inode Number	2151633962
Access Permission	rwxr-xr-x
Owner Name	
File Size	0KB
File Size (bytes)	90

#	Name	Member Type	File Size	Last Modified
1	FORMATS	CATALOG	24KB	09/12/2022 18:34:45
2	FORMATS_EN_CA	CATALOG	24KB	09/14/2022 00:10:00
3	FORMATS_FR_CA	CATALOG	24KB	09/14/2022 00:10:00

What formats are in my format library?

```
proc catalog catalog=CARlib.formats;  
    contents;  
quit;
```

Contents of Catalog CARLIB.FORMATS					
#	Name	Type	Create Date	Modified Date	Description
1	ENGFMT	FORMAT	09/13/2022 17:42:52	09/13/2022 17:42:52	
2	MSRPFMT	FORMAT	09/13/2022 17:42:17	09/13/2022 17:42:17	

What's in a format?

```
proc format library=CARlib cntlout=work.MSRPfmt_desc;
  select MSRPfmt;
```

```
run;
```

```
proc print data=work.MSRPfmt_desc;
```

```
run;
```

Obs	FMTNAME	START	END	LABEL	MIN	MAX	DEFAULT	LENGTH	FUZZ	PREF
1	MSRPFMT	LOW	10000	Less than \$10,000	1	40	20	20	1E-12	
2	MSRPFMT	10000	20000	\$10,000 to \$20,000	1	40	20	20	1E-12	
3	MSRPFMT	20000	30000	\$20,000 to \$30,000	1	40	20	20	1E-12	
4	MSRPFMT	30000	40000	\$30,000 to \$40,000	1	40	20	20	1E-12	
5	MSRPFMT	40000	50000	\$40,000 to \$50,000	1	40	20	20	1E-12	
6	MSRPFMT	50000	60000	\$50,000 to \$60,000	1	40	20	20	1E-12	
7	MSRPFMT	60000	HIGH	Greater than \$60,000	1	40	20	20	1E-12	

Create a format from a SAS dataset

```
proc format library=CARlib cntlin=work.MSRPfmt_desc;  
run;
```

```
--  
71      proc format library=CARlib cntlin=work.MSRPfmt_desc;  
NOTE: Format MSRPFMT is already on the library CARLIB.FORMATS.  
NOTE: Format MSRPFMT has been written to CARLIB.FORMATS.  
72      run;
```

```
NOTE: PROCEDURE FORMAT used (Total process time):  
      real time           0.00 seconds  
      user cpu time       0.00 seconds  
      system cpu time     0.00 seconds  
      memory              590.25k  
      OS Memory           22444.00k  
      Timestamp           09/14/2022 12:51:36 AM  
      Step Count          100  Switch Count  0  
      Page Faults         0
```

What libraries will be searched for formats?

```
proc options option=fmtsearch;  
run;
```

```
70  
71      proc options option=fmtsearch;  
72      run;
```

SAS (r) Proprietary Software Release 9.4 TS1M6

FMTSEARCH=(APFMTLIB WORK LIBRARY)

Specifies the order in which format catalogs are searched.

NOTE: PROCEDURE OPTIONS used (Total process time):

1.11 0.00 0.00 0.00

Changing formats based on the locale value!



Setup

```
libname langtst "/home/xxx/wuss";  
run;
```

```
proc format lib=langtst.formats_en_CA;  
    value lngfmt  
        1="One"  
        2="Two";  
run;
```

```
proc format lib=langtst.formats_fr_CA;  
    value lngfmt  
        1="Un"  
        2="Deux";  
run;
```

```
options fmtsearch=(langtst /locale);  
run;
```

Change formats based on the "LOCALE" value

```
data mynums;  
    format printnum lngfmt.;  
    input printnum;  
    cards;
```

```
1  
2  
3  
run;
```

```
options locale="fr_CA";  
run;
```

```
proc print data=mynums;  
run;
```

Obs.	printnum
1	Un
2	Deux
3	3

```
options locale="en_CA";
```

```
proc print data=mynums;  
run;
```

Obs	printnum
1	One
2	Two
3	3

Contact Information



Name: Tom Kari

Company: Tom Kari Consulting

City/State: Ottawa, Canada

Phone: 1-613-899-6359

Email: tom@tomkari.ca